

A VERY FAST, VERY STRONG, TOPOLOGICALLY MEANINGFUL AND FUN KNOT INVARIANT

DROR BAR-NATAN AND ROLAND VAN DER VEEN

ABSTRACT. In this paper we introduce $\Theta = (\Delta, \theta)$, a pair of polynomial knot invariants which is:

- Theoretically and practically fast: Θ can be computed in polynomial time and we computed it in full on random knots with over 300 crossings, and its evaluation on simple rational numbers on random knots with over 700 crossings.
- Strong: Its separation power is much greater than, say, the HOMFLY-PT polynomial and Khovanov homology (taken together) on knots with up to 15 crossings (while computing much faster).
- Topologically meaningful: It gives a genus bound, and there are reasons to hope that it would do more.
- Fun: Scroll to Figures 1.1, 1.2, and 3.1.

Δ is merely the Alexander polynomial. θ is almost certainly equal to an invariant that was studied extensively by Ohtsuki [Oh], continuing Rozansky, Garoufalidis, and Kricker [GR, Ro1, Ro2, Ro3, Kr]. Yet our formulas, proofs, and programs are much simpler and enable its computation even on very large knots.

CONTENTS

1. Fun	3
2. Formulas	7
2.1. Old Formulas	7
2.2. New Formulas	8
3. Implementation and Examples	11
4. Proof of Invariance	15
5. Strong and Meaningful	23
6. Conjectures and Dreams	23
References	23

Date: First edition Not Yet. This edition July 10, 2025.


2020 Mathematics Subject Classification. Primary 57K14, secondary 16T99.

Key words and phrases. Alexander polynomial, TBW .

This work was partially supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC). It is available in electronic form, along with source files and a demo *Mathematica* notebook at <http://drorbn.net/Theta> and at [arXiv:2507.06111](https://arxiv.org/abs/2507.06111).


1. FUN

The word “fun” rarely appears in the title of a math paper, so let us start with a brief justification.

Θ is a pair of polynomials. The first, Δ , is old news, the Alexander polynomial [Al]. It is a one-variable Laurent polynomial in a variable T . For example, $\Delta(\textcircled{3}) = T^{-1} - 1 + T$. We turn such a polynomial to a list of coefficients (for $\textcircled{3}$, it is $(1 \ -1 \ 1)$), and then to a chain of bars of varying colours: white for the zero coefficients, and red and blue for the positive and negative coefficients (with intensity proportional to the magnitude of the coefficients). The result is a “bar code”, and for the trefoil $\textcircled{3}$ is it .

Similarly, θ is a 2-variable Laurent polynomial, in variables T_1 and T_2 . We can turn such a polynomial into a 2D array of coefficients and then using the same rules, into a 2D array of colours, namely into a picture! To highlight a certain conjectured hexagonal symmetry of the resulting pictures, we apply a certain shear transformation to the plane before printing. So the colour of a monomial $cT_1^{n_1}T_2^{n_2}$ gets printed at position

$\begin{pmatrix} 1 & -1/2 \\ 0 & \sqrt{3}/2 \end{pmatrix} \begin{pmatrix} n_1 \\ n_2 \end{pmatrix}$ instead of the more traditional $\begin{pmatrix} n_1 \\ n_2 \end{pmatrix}$. On the right is the 2D picture corresponding to the polynomial $2 + T_1 - T_1T_2 + T_2 - T_1^{-1} + T_1^{-1}T_2^{-1} - T_2^{-1}$.




Thus Θ becomes a pair of pictures: a bar code, and a 2D picture that we call a “hexagonal QR code”. For the knots in the Rolfsen table (with the unknot prepended at the start), they are in Figure 1.1. In addition, the hexagonal QR codes of some 15 knots with ≥ 300 crossings are in Figure 1.2, and Θ of a 132-crossing torus knot is in Figure 3.1.

MORE. Add some mats.

Clearly there are patterns in these figures. There is a hexagonal symmetry and the QR codes are nearly always hexagons (these are independent properties). Much more can be seen in Figure 1.1. In Figure 1.2 there seem to be large-scale “sand table patterns” or “diffraction patterns”. We can’t prove any of these things, and the last one, we can’t even formulate properly. Yet they are clearly there, too clear to be the result of chance alone.

We plan to have fun over the next few years observing and proving these patterns. We hope that others will join us too.

FIGURE 1.1. Θ as a bar code and a QR code, for all the knots in the Rolfsen table.




FIGURE 1.2. θ (hexagonal QR code only) of the 15 largest knots that we have computed by September 16, 2024. They are all “generic” in as much as we know, and they all have ≥ 300 crossings. The knots come from [DHOEBL]. Warning: Some screens/printers may display spurious Moiré interference patterns.


fig:300

2. FORMULAS

2.1. Old Formulas¹. The setup leading to the definition of Θ is the same as the setup leading to the definition of the invariant ρ_1 of [BV1], and hence we copy a few relevant paragraphs from [BV1] nearly varbatim, with only a few modifications.

Given an oriented n -crossing knot K , we draw it in the plane as a long knot diagram D in such a way that the two strands intersecting at each crossing are pointing up (that's always possible because we can always rotate crossings as needed), and so that at its beginning and at its end the knot is oriented upward. We call such a diagram an *upright knot diagram*. An example of an upright knot diagram is shown on the right.

We then label each edge of the diagram with two labels: a running index k which runs from 1 to $2n + 1$, and a “rotation number” φ_k , the geometric rotation number of that edge (the signed number of times the tangent to the edge is horizontal and heading right, with cups counted with $+1$ signs and caps with -1 ; this number is well defined because at their ends, all edges are headed up). On the right the running index runs from 1 to 7, and the rotation numbers for all edges are 0 (and hence are omitted) except for φ_4 , which is -1 .



Technicality 1. Some Reidemeister moves create or lose an edge and to avoid the need for renumbering it is beneficial to also allow labelling the edges with non-consecutive labels. Hence we allow that, and write i^+ for the successor of the label i along the knot, and i^{++} for the successor of i^+ (these are $i + 1$ and $i + 2$ if the labelling is by consecutive integers). Also, by convention “1” will always refer to the label of the first edge, and “ $2n + 1$ ” will always refer to the label of the last. 1

Let X be the set of all crossings in the diagram D , where we encode each crossing as a triple (sign, incoming over edge, incoming under edge). In our example we have $X = \{(1, 1, 4), (1, 5, 2), (1, 3, 6)\}$.

We let A be the $(2n + 1) \times (2n + 1)$ matrix of Laurent polynomials in a formal variable T , defined by

$$A := I - \sum_c (T^s E_{i,i^+} + (1 - T^s) E_{i,j^+} + E_{j,j^+}),$$

where I is the identity matrix and $E_{\alpha\beta}$ denotes the elementary matrix with 1 in row α and column β and zeros elsewhere. The summation is over the crossings $c = (s, i, j)$ of the diagram D , and once c is chosen, s denotes its sign and i and j denote the labels below the crossing where the label i belongs to the over-strand and j to the under-strand.

Alternatively, $A = I + \sum_c A_c$, where A_c is a matrix of zeros except for the blocks as follows:

$$\begin{array}{c} \begin{array}{c} i^+ \uparrow \quad i^+ \uparrow \\ \diagdown \quad \diagup \\ i \quad j \\ \diagup \quad \diagdown \\ j \quad i \\ s = +1 \end{array} \quad \begin{array}{c} i^+ \uparrow \quad j^+ \uparrow \\ \diagdown \quad \diagup \\ j \quad i \\ \diagup \quad \diagdown \\ i \quad j^+ \\ s = -1 \end{array} \end{array} \longrightarrow \begin{array}{c|cc} A_c & \text{column } i^+ & \text{column } j^+ \\ \hline \text{row } i & -T^s & T^s - 1 \\ \text{row } j & 0 & -1 \end{array} \quad (1)$$

¹“Old” means that these formulas appeared already in [BV1].

We note (as we did in [BV1]) that the determinant of A is equal up to a unit to the normalized Alexander polynomial Δ of K . In fact, we have that

$$\Delta = T^{(-\varphi(D)-w(D))/2} \det(A), \quad (2) \quad \text{eq:Delta}$$

where $\varphi(D) := \sum_k \varphi_k$ is the total rotation number of D and where $w(D) = \sum_c s_c$ is the writhe of D , namely the sum of the signs s_c of all the crossings c in D .

We let $G = (g_{\alpha\beta}) = A^{-1}$ and, thinking of it as a function $g_{\alpha\beta}$ of a pair of edges α and β , we call it the Green function of the diagram D . When inspired by physics (e.g. [BN2]), we sometimes call it “the 2-point function”, and when thinking of car traffic (e.g. [BN3]) we sometimes call it “the traffic function”.

We note that the computation of G is the bottleneck in the computation of Θ . It requires inverting a $(2n+1) \times (2n+1)$ matrix whose entries are (degree 1) Laurent polynomials in T . It’s a daunting task yet it takes polynomial time, it can be performed in practice even if n is in the hundreds, and everything which then follows is easier.

2.2. New Formulas. Let T_1 and T_2 be indeterminates and let $T_3 := T_1 T_2$. Let $\Delta_\nu := \Delta_{T \rightarrow T_\nu}$ and $G_\nu = (g_{\nu\alpha\beta}) := G_{T \rightarrow T_\nu}$ be Δ and G subject to the substitution $T \rightarrow T_\nu$, where $\nu = 1, 2, 3$ (these are easy to compute once Δ and G have been computed).

Given crossings $c = (s, i, j)$, $c_0 = (s_0, i_0, j_0)$, and $c_1 = (s_1, i_1, j_1)$ in X , let

$$F_1(c) = s [1/2 - g_{3ii} + T_2^s g_{1ii} g_{2ji} - T_2^s g_{3jj} g_{2ji} - (T_2^s - 1) g_{3ii} g_{2ji} \quad (3) \quad \text{eq:F1}$$

$$+ (T_3^s - 1) g_{2ji} g_{3ji} - g_{1ii} g_{2jj} + 2 g_{3ii} g_{2jj} + g_{1ii} g_{3jj} - g_{2ii} g_{3jj}]$$

$$+ \frac{s}{T_2^s - 1} [(T_1^s - 1) T_2^s (g_{3jj} g_{1ji} - g_{2jj} g_{1ji} + T_2^s g_{1ji} g_{2ji})$$

$$+ (T_3^s - 1) g_{3ji} (1 - T_2^s g_{1ii} + g_{2ij} + (T_2^s - 2) g_{2jj} - (T_1^s - 1)(T_2^s + 1) g_{1ji})]$$


$$F_2(c_0, c_1) = \frac{s_1 (T_1^{s_0} - 1) (T_3^{s_1} - 1) g_{1j_1 i_0} g_{3j_0 i_1}}{T_2^{s_1} - 1} (T_2^{s_0} g_{2i_1 i_0} + g_{2j_1 j_0} - T_2^{s_0} g_{2j_1 i_0} - g_{2i_1 j_0}) \quad (4) \quad \text{eq:F2}$$

$$F_3(\varphi, k) = \varphi (g_{3kk} - 1/2) \quad (5) \quad \text{eq:F3}$$

Theorem 2 (Proof in Section 4). *The following is a knot invariant:*

$$\theta(D) := \Delta_1 \Delta_2 \Delta_3 \left(\sum_{c \in X} F_1(c) + \sum_{c_0, c_1 \in X} F_2(c_0, c_1) + \sum_{\text{edges } k} F_3(\varphi_k, k) \right). \quad (6) \quad \text{eq:Main}$$

We note without detail that there is an alternative formula for θ in terms of perturbed Gaussian integration [BN2]. In that language, and using also the traffic motifs of [BV1, BN3], the three summands in (6) become Feynman diagrams for processes in which cars governed by parameter $T = T_1, T_2$, or T_3 interact:



In particular, the middle diagram which resembles the greek letter Θ gave the invariant its name.



We note also that computationally, the worst term in (6) is the middle one, and even it takes merely $\sim n^2$ operations in the ring $\mathbb{Q}(T_1, T_2)$ to complete.

The polynomials $F_1(c)$, $F_2(c_0, c_1)$ and $F_3(\varphi, k)$ are not unique, and we are not certain that we have the cleanest possible formulas for them. They are human-ugly, yet from a computational perspective, having 18 terms (as is the case for $F_1(c)$) isn't really a problem; computers don't care.

3. IMPLEMENTATION AND EXAMPLES



A concise yet reasonably efficient implementation is worth a thousand formulas. It completely removes ambiguities, it tests the theories, and it allows for experimentation. Hence our next task is to implement. The section that follows was generated from a Mathematica [Wo] notebook which is available at [BV2, Theta.nb]. A second implementation of Θ , using Python and SageMath (<https://www.sagemath.org/>) is available at <https://www.rolandvdv.nl/Theta/>.

We start by loading the package `KnotTheory` — it is only needed because it has many specific knots pre-defined:

 `<< KnotTheory``  Loading KnotTheory` version of October 29, 2024, 10:29:52.1301.
Read more at <http://katlas.org/wiki/KnotTheory>.



Next we quietly define the modules `Rot`, used to compute rotation numbers, and `PolyPlot`, used to plot polynomials as bar codes and as hexagonal QR codes. Neither is a part of the core of the computation of Θ , so neither is shown; yet we do show one usage example for each.


 `(* Rot suppressed *)`

 `Rot[Mirror@Knot[3, 1]]`  `{{{1, 1, 4}, {1, 3, 6}, {1, 5, 2}}, {0, 0, 0, -1, 0, 0, 0}}`

We urge the reader to compare the above output with the knot diagram in Section 2.1.

 `(* PolyPlot suppressed *)`


 `PolyPlot[{2 T - 1 + T-1, -1 + T1 - 2 T2 + 4 T1-1 T2-1},`
 `ImageSize → 100, Labeled → True]`







The definition of `CF` below is a technicality telling the computer how to best store polynomials in the $g_{\nu\alpha\beta}$ such as F_1 and F_2 . The programs would run just the same without it, albeit a bit more slowly:


 `CF[\mathcal{E}] := Expand@Collect[\mathcal{E} , $g_{_}$, F] /. F → Factor;`

Next, we decree that $T_3 = T_1 T_2$ and define the three “Feynman Diagram” polynomials F_1 , F_2 , and F_3 :



 `T3 = T1 T2;`

 `F1[{ $s_{_}$, $i_{_}$, $j_{_}$ }] := CF[`
 `$s \left(\frac{1}{2} - g_{3ii} + T_2^s g_{1ii} g_{2ji} - g_{1ii} g_{2jj} - (T_2^s - 1) g_{2ji} g_{3ii} + 2 g_{2jj} g_{3ii} - (1 - T_3^s) g_{2ji} g_{3ji} - \right.$`
 `$g_{2ii} g_{3jj} - T_2^s g_{2ji} g_{3jj} + g_{1ii} g_{3jj} +$`
 `$((T_1^s - 1) g_{1ji} (T_2^s g_{2ji} - T_2^s g_{2jj} + T_2^s g_{3jj}) +$`
 `$(T_3^s - 1) g_{3ji} (1 - T_2^s g_{1ii} + g_{2ij} + (T_2^s - 2) g_{2jj} - (T_1^s - 1) (T_2^s + 1) g_{1ji})) / (T_2^s - 1)]$`


 $F_2[\{s_, i_, j_ \}, \{s1_, i1_, j1_ \}] :=$
 $CF[s1 (T_1^{s\theta} - 1) (T_2^{s1} - 1)^{-1} (T_3^{s1} - 1) g_{1,j1,i\theta} g_{3,j\theta,i1}$
 $((T_2^{s\theta} g_{2,i1,i\theta} - g_{2,i1,j\theta}) - (T_2^{s\theta} g_{2,j1,i\theta} - g_{2,j1,j\theta}))]$


 $F_3[\varphi_, k_] = \varphi g_{3kk} - \varphi / 2;$


Next comes the main program computing Θ . Fortunately, it matches perfectly with the mathematical description in Section 2. In line 01 we let X be the list of crossings in an input knot K , and φ the list of its rotation numbers, using the external program `Rot` which we have already mentioned. We also let n be the length of X , namely, the number of crossings in K . In line 02 we let the starting value of A be the identity matrix, and then in line 03, for each crossing in X we add to A a 2×2 block, in rows i and j and columns $i + 1$ and $j + 1$, as explain in Equation (1). In line 04 we compute the normalized Alexander polynomial Δ as in (2). In line 05 we let G be the inverse of A . In line 06 we declare what it means to evaluate, `ev`, a formula \mathcal{E} that may contain symbols of the form $g_{\nu\alpha\beta}$: each such symbol is to be replaced by the entry in position α, β of G , but with T replaced with T_ν . In line 07 we start computing θ by computing the first summand in (6), which in itself, is a sum over the crossings of the knot. In line 08 we add to θ the double sum corresponding to the second term in (6), and in line 09, we add the third summand of (6). Finally, line 10 outputs a pair: Δ , and the re-normalized version of θ .


 $\Theta[K_]:= \Theta[K] = \text{Module}[\{X, \varphi, n, A, \Delta, G, \text{ev}, \theta\},$
 $(*$ 01 $*) \{X, \varphi\} = \text{Rot}[K]; n = \text{Length}[X];$
 $(*$ 02 $*) A = \text{IdentityMatrix}[2 n + 1];$
 $(*$ 03 $*) \text{Cases}[X, \{s_, i_, j_ \} \Rightarrow (A[[\{i, j\}, \{i + 1, j + 1\}]] += \begin{pmatrix} -T^s & T^s - 1 \\ \theta & -1 \end{pmatrix})];$
 $(*$ 04 $*) \Delta = T^{(-\text{Total}[\varphi] - \text{Total}[X[[\text{All}, 1]]]) / 2} \text{Det}[A];$
 $(*$ 05 $*) G = \text{Inverse}[A];$
 $(*$ 06 $*) \text{ev}[\mathcal{E}_] := \text{Factor}[\mathcal{E} /. g_{\nu_, \alpha_, \beta_] \Rightarrow (G[[\alpha, \beta]] /. T \rightarrow T_\nu)];$
 $(*$ 07 $*) \theta = \text{ev}[\sum_{k=1}^n F_1[X[[k]]];$
 $(*$ 08 $*) \theta += \text{ev}[\sum_{k1=1}^n \sum_{k2=1}^n F_2[X[[k1]], X[[k2]]];$
 $(*$ 09 $*) \theta += \text{ev}[\sum_{k=1}^{2n} F_3[\varphi[[k]], k];$
 $(*$ 10 $*) \text{Factor}@\{\Delta, (\Delta /. T \rightarrow T_1) (\Delta /. T \rightarrow T_2) (\Delta /. T \rightarrow T_3) \theta\}$
 $];$

On to examples! Starting with the trefoil knot.


 $\text{Expand}[\Theta[\text{Knot}[3, 1]]]$


 $\left\{ -1 + \frac{1}{T} + T, -\frac{1}{T_1^2} - T_1^2 - \frac{1}{T_2^2} - \frac{1}{T_1^2 T_2^2} + \frac{1}{T_1 T_2^2} + \frac{1}{T_1^2 T_2} + \frac{T_1}{T_2} + \frac{T_2}{T_1} + T_1^2 T_2 - T_2^2 + T_1 T_2^2 - T_1^2 T_2^2 \right\}$


 `PolyPlot[Θ [Knot[3, 1]], ImageSize \rightarrow Tiny]`




Next are the Conway knot 11_{n34} and the Kinoshita-Terasaka knot 11_{n42} . The two are mutants and famously hard to separate: they both have $\Delta = 1$ (as evidenced by their one-bar bar codes below), and they have the same HOMFLY-PT polynomial and Khovanov homology. Yet their θ invariants are different. Note that the genus of the Conway knot is 3, while the genus of the Kinoshita-Terasaka knot is 2. This agrees with the apparent higher complexity of the QR code of the Conway polynomial, and with the observations in Section 5.




 `PolyPlot[Θ [Knot[#]], ImageSize \rightarrow 120] & /@ {"K11n34", "K11n42"}`




Torus knots have particularly nice-looking Θ invariants. Here are the torus knots $T_{13/2}$, $T_{17/3}$, $T_{13/5}$, and $T_{7/6}$:


 `GraphicsRow[ImageCompose[
PolyPlot[Θ [TorusKnot@@ #], ImageSize \rightarrow 480],
TubePlot[TorusKnot@@ #, ImageSize \rightarrow 240],
{Right, Bottom}, {Right, Bottom}
] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}]`



The next line shows the computation time in seconds for the 132-crossing torus knot $T_{22/7}$ on a 2024 laptop, without actually showing the output. The output plot is in Figure 3.1.

 `AbsoluteTiming[Θ [TorusKnot[22, 7]]];]`

 {1020.73, Null}

 `ImageCompose[PolyPlot[Θ [TorusKnot[22, 7]], ImageSize \rightarrow 720],
TubePlot[TorusKnot[22, 7], ImageSize \rightarrow 360], {Right, Bottom}, {Right, Bottom}]`




FIGURE 3.1. The 132-crossing torus knot $T_{22/7}$ and a plot of its Θ invariant

4. PROOF OF INVARIANCE

Our proof of the invariance of θ (Theorem 2) is very similar, and uses many of the same pieces, as the proof of the invariance of ρ_1 in [BV1]. Thus instead of repeating everything we just summarize those steps that are identical and then provide the needed details for the steps that differ.

Like in [BV1, Lemma 3], the equalities $AG = I$ and $GA = I$ imply that for any crossing $c = (s, i, j)$ in a knot diagram D , the Green function $G = (g_{\alpha\beta})$ of D satisfies the following “ g -rules”, with δ denoting the Kronecker delta:

$$g_{i\beta} = \delta_{i\beta} + T^s g_{i^+, \beta} + (1 - T^s) g_{j^+, \beta}, \quad g_{j\beta} = \delta_{j\beta} + g_{j^+, \beta}, \quad g_{2n+1, \beta} = \delta_{2n+1, \beta}, \quad (7)$$

$$g_{\alpha, i^+} = T^s g_{\alpha i} + \delta_{\alpha, i^+}, \quad g_{\alpha, j^+} = g_{\alpha j} + (1 - T^s) g_{\alpha i} + \delta_{\alpha, j^+}, \quad g_{\alpha, 1} = \delta_{\alpha, 1}. \quad (8)$$

Furthermore, the systems of equations (7) is equivalent to $AG = I$ and so it fully determines $g_{\alpha\beta}$, and likewise for the system (8), which is equivalent to $AG = I$.

Of course, the same g -rules also hold for $G_\nu = (g_{\nu\alpha\beta})$ for $\nu = 1, 2, 3$, except with T replaced with T_ν .


We also need a variant \tilde{g}_{ab} of $g_{\alpha\beta}$, defined whenever a and b are two distinct points on the edges of a knot diagram D , away from the crossings. If α is the edge on which a lies and β is the edge on which b lies, \tilde{g}_{ab} is defined as follows:

$$\tilde{g}_{ab} = \begin{cases} g_{\alpha\beta} & \text{if } \alpha \neq \beta, \\ g_{\alpha\beta} & \text{if } \alpha = \beta \text{ and } a < b \text{ relative to the orientation of the edge } \alpha = \beta, \\ g_{\alpha\beta} - 1 & \text{if } \alpha = \beta \text{ and } a > b \text{ relative to the orientation of the edge } \alpha = \beta. \end{cases} \quad (9)$$

Of course, we can define $\tilde{g}_{\nu ab}$ from $g_{\alpha\beta}$ in an identical way.

It is clear that g and \tilde{g} contain the same information and are easily computable from each other. The variant \tilde{g} is, strictly speaking, not a matrix and so g is a bit more suitable for computations. Yet \tilde{g} is a bit better behaved when we try to track, as below, the behaviour of g / \tilde{g} under Reidemeister moves. Reidemeister moves sometimes merge two edges into one or break an edge into two. In such cases the points a and b can be “pulled” along with the move so as to retain their ordering along the overall parametrization of the knot, yet mere edge labels lose this information. The following discussion and lemma exemplify the advantage of \tilde{g} of g :

Discussion 3. We introduce “null vertices” as on the right into knot diagrams, whose only function (as we shall see) is to cut edges into parts that may carry different labels. The matrix A becomes a bit larger (by as many null vertices as were added to a knot diagram). The rule (1) for the creation of the matrix A gets an amendment for null vertices,



$$\begin{array}{c} j \quad k \\ \text{---} \bullet \text{---} \end{array} \longrightarrow \begin{array}{c|c} A_{nv} & \text{column } k \\ \hline \text{row } j & -1 \end{array},$$

and the summation for A , $A = I + \sum_c A_c + \sum_{nv} A_{nv}$ is extended to include summands for the null vertices. The matrix $G = A^{-1}$ and the function $g_{\alpha\beta}$ are defined as before. The g -rules of (7) and (8) get additions,




FIGURE 4.1. The modified Green function \tilde{g}_{ab} is invariant under Reidemeister moves performed away from where it is measured.

$$g_{j\beta} = \delta_{j\beta} + g_{k\beta},$$

(10)

eq:NullCircles

$$g_{\alpha k} = \delta_{\alpha k} + g_{\alpha j},$$

(11)

eq:NullCircles

and it remains true that the system of equations (7) \cup (10) (as well as (8) \cup (11)) fully determines $g_{\alpha\beta}$. The variant \tilde{g}_{ab} is also defined as before, except now a and b also need to be away from the null vertices.

Lemma 4. *Inserting a null vertex does not change \tilde{g}_{ab} , provided it is inserted away from a and b . (This statement does not make sense for $g_{\alpha\beta}$, as inserting a null vertex changes the dimensions of the matrix $G = (g_{\alpha\beta})$).*

Proof. Let D be an upright knot diagram having an edge labeled i and let D' be obtained from it by adding a null vertex within edge i , naming the two resulting half-edges j and k (in order). Let $g_{\alpha\beta}$ be the Green function for D , and similarly, $g'_{\alpha\beta}$ for D' . We claim that

$$g'_{\alpha\beta} = \begin{array}{c|ccc} & \beta = j & \beta = k & \beta \notin \{j, k\} \\ \hline \alpha = j & g_{ii} & g_{ii} & g_{i\beta} \\ \alpha = k & g_{ii} - 1 & g_{ii} & g_{i\beta} \\ \alpha \notin \{j, k\} & g_{\alpha i} & g_{\alpha i} & g_{\alpha\beta} \end{array}.$$

Indeed, all we have to do is to verify that the above-defined $g'_{\alpha\beta}$ satisfies all the g -rules (7) \cup (10), and that is easy. The lemma now follows easily from the definition of \tilde{g}' in Equation (9). \square

Remark 5. The statement of Theorem 2 does not change in the presence of null vertices: There are no “ F ” terms for those, and their only effect on the definition of Θ in Equation (6) is to change the edge labels that appear within c , c_1 , and c_2 , and within the F_3 sum.

The following theorem, was not named in [BV1], yet it was stated there as the first part of the first proof of [BV1, Theorem 1].

Theorem 6. *The variant Green function \tilde{g}_{ab} is a “relative invariant”, meaning that once points a and b are fixed within a knot diagram D , the value of \tilde{g}_{ab} does not change if Reidemeister moves are performed away from the points a and b . An illustration appears in Figure 4.1. It follows that the same is also true for $\tilde{g}_{\nu ab}$ for $\nu = 1, 2, 3$.*


We note that \tilde{g}_{ab} is nearly the same as $g_{\alpha\beta}$, if a is on α and b is on β . So Theorem 6 also says that $g_{\alpha\beta}$ is invariant under Reidemeister moves away from α and β , except for edge-renumbering issues and ± 1 contributions that arise if α and β correspond to edge that get merged or broken by the Reidemeister moves.

The proof of Theorem 6 is perhaps best understood in terms of the traffic function of [BV1, BN1, BN3]: One simply needs to verify that for each of the Reidemeister moves, traffic entering the tangle diagram for the left hand side of the move exits it in the same manner

Fig: A ✓ spanning set of oriented Reidemeister moves as in - - -

as traffic entering the tangle diagram for the right hand side of the move, and each of these verifications, as explained in [BV1, BN1, BN3], is very easy. Yet that proof is a bit informal, so we opt here to give a fully formal proof along the lines of the first halves of [BV1, Propositions 7-9].

Proof of Theorem 6. We need to know how the Green function $g_{\alpha\beta}$ changes under Reidemeister moves, and we start with R3. Below are the two sides of the move, along with the g -rules of type (7) corresponding to the crossings within, written with the assumption that β isn't in $\{i^+, j^+, k^+\}$, so several of the Kronecker deltas can be ignored. We use g for the Green function at the left-hand side of R3, and g' for the right-hand side, and recall that along with the further g/g' -rules corresponding to all the non-moving knot crossings, these rules fully determine $g_{\alpha\beta}$ and $g'_{\alpha\beta}$ for $\beta \notin \{i^+, j^+, k^+\}$:



A routine computation (eliminating $g_{i+, \beta}$, $g_{j+, \beta}$, and $g_{k+, \beta}$) shows that the first system of 6 equations is equivalent to the following system of 6 equations:

$$\begin{aligned}
 g_{i, \beta} &= \delta_{i\beta} + T^2 g_{i++, \beta} + T(1-T) g_{j++, \beta} + (1-T) g_{k++, \beta}, \\
 g_{j, \beta} &= \delta_{j\beta} + T g_{j++, \beta} + (1-T) g_{k++, \beta}, \quad g_{k, \beta} = \delta_{k\beta} + g_{k++, \beta},
 \end{aligned} \tag{12}$$

$$g_{i+, \beta} = T g_{i++, \beta} + (1-T) g_{j++, \beta}, \quad g_{j+, \beta} = g_{j++, \beta}, \quad g_{k+, \beta} = g_{k++, \beta}. \tag{13}$$

In this system the indices i^+ , j^+ and k^+ do not appear in (12) or in the further g -rules corresponding to the further crossings. Hence for the purpose of determining $g_{\alpha\beta}$ with $\alpha, \beta \notin \{i^+, j^+, k^+\}$, Equations (13) can be ignored.

Similarly eliminating $g'_{i+, \beta}$, $g'_{j+, \beta}$, and $g'_{k+, \beta}$ from the second set of equations, we find that it is equivalent to

$$\begin{aligned}
 g'_{i, \beta} &= \delta_{i\beta} + T^2 g'_{i++, \beta} + T(1-T) g'_{j++, \beta} + (1-T) g'_{k++, \beta}, \\
 g'_{j, \beta} &= \delta_{j\beta} + T g'_{j++, \beta} + (1-T) g'_{k++, \beta}, \quad g'_{k, \beta} = \delta_{k\beta} + g'_{k++, \beta},
 \end{aligned} \tag{14}$$

$$g'_{i+, \beta} = T g'_{i++, \beta} + (1-T) g'_{k++, \beta}, \quad g'_{j+, \beta} = T g'_{j++, \beta} + (1-T) g'_{k++, \beta}, \quad g'_{k+, \beta} = g'_{k++, \beta}. \tag{15}$$

Using the same logic as before, for the purpose of determining $g'_{\alpha\beta}$ with $\alpha, \beta \notin \{i^+, j^+, k^+\}$, Equations (15) can be ignored.

But now we compare the unignored equations, (12) and (14) and find that they are exactly the same, except with $g \leftrightarrow g'$, and the same is true for the further g/g' -rules coming from the further crossings. Hence so long as $\alpha, \beta \notin \{i^+, j^+, k^+\}$, we have that $g_{\alpha\beta} = g'_{\alpha\beta}$. In the case of the R3 move no edges merge or break up, and hence this implies that $\tilde{g}_{ab} = \tilde{g}'_{ab}$ so long as a and b are away from the move.




FIGURE 4.2. The upright Reidemeister moves: Reidemeister 1 left and right, Reidemeister 2 braid-like and cyclic, Reidemeister 3, and (the +) Swirl.

Next we deal with the case of cyclic R2 moves – R2 moves in which the two strands involved have opposing orientations. We use the privileges afforded to us by Lemma 4 to insert 4 null vertices into the right-hand-side of the R2c moves, and like in the case of R3, we start with pictures annotated with the relevant type (7) and (10) g -rules, written with the assumption that $\beta \notin \{i^+, j^+\}$:

wn-upright.

	$g_{i^+, \beta} = T g_{i^+, \beta} + (1 - T) g_{j^+, \beta}$ $g_{j, \beta} = \delta_{j, \beta} + g_{j^+, \beta}$ $g_{i, \beta} = \delta_{i, \beta} + T^{-1} g_{i^+, \beta} + (1 - T^{-1}) g_{j^+, \beta}$ $g_{j^+, \beta} = g_{j^+, \beta}$		$g'_{i, \beta} = \delta_{i, \beta} + g'_{i^+, \beta}$ $g'_{j^+, \beta} = g'_{j^+, \beta}$ $g'_{i^+, \beta} = g'_{i^+, \beta}$ $g'_{j, \beta} = \delta_{j, \beta} + g'_{j^+, \beta}$
...		...	
further crossings	further g -rules	further crossings	further g' -rules

As in the case of R3, we eliminate $g_{i^+, \beta}$ and $g_{j^+, \beta}$ from the equations for the left hand side, and find that for the purpose of determining $g_{\alpha\beta}$ with $\beta \notin \{i^+, j^+\}$, they are equivalent to the equations

$$g_{i, \beta} = \delta_{i, \beta} + g_{i^+, \beta} \quad \text{and} \quad g_{j, \beta} = \delta_{j, \beta} + g_{j^+, \beta}.$$

Likewise, the right hand side is clearly equivalent to

$$g'_{i, \beta} = \delta_{i, \beta} + g'_{i^+, \beta} \quad \text{and} \quad g'_{j, \beta} = \delta_{j, \beta} + g'_{j^+, \beta},$$

and as in the case of R3, this establishes the invariance of \tilde{g}_{ab} under R2c moves.

MORE.

We can now move on to the main part of the proof of Theorem 2. As follows from [BV1, Theorem 2], we need to prove the invariance of θ under the “upright Reidemeister” moves of Figure 4.2. We start with the hardest, R3:

prop:R3

Proposition 7. *The quantity θ is invariant under R3.*

Proof. Let D_l and D_r be two knot diagrams that differ only by an R3 move, and label their relevant edges and crossings as in Figure 4.3. Let $g_{\nu\alpha\beta}^l$ and $g_{\nu\alpha\beta}^r$ be their corresponding Green functions. Let $F_1^l(c)$, $F_2^l(c_0, c_1)$ and $F_3^l(\varphi, k)$ be defined from $g_{\nu\alpha\beta}^l$ as in (3)–(5), and similarly make F_1^r , F_2^r and F_3^r using $g_{\nu\alpha\beta}^r$.




FIGURE 4.3. The two sides D^l and D^r of the R3 move. The left side D^l consists of 3 distinguished crossings $c_1^l = (1, j, k)$, $c_2^l = (1, i, k^+)$, $c_3^l = (1, i^+, j^+)$ and a collection of further crossings $c_y = (s, m, n) \in Y$, where Y is the set of crossings not participating in the R3 move. The right side D^r consists of $c_1^r = (1, i, j)$, $c_2^r = (1, i^+, k)$, $c_3^r = (1, j^+, k^+)$ and the same set Y of further crossings c_y .

fig:R3

By the invariance of the Alexander polynomial, the pre-factor $\Delta_1 \Delta_2 \Delta_3$ is the same for $\theta(D^l)$ and for $\theta(D^r)$ (see Equation (6)). By Theorem 6, $g_{\nu\alpha\beta}^l = g_{\nu\alpha\beta}^r$ so long as $\alpha, \beta \notin \{i^+, j^+, k^+\}$. And so the only terms that may differ in $\theta(D^h)$ between $h = l$ and $h = r$ are the terms

$$A^h = \sum_{c \in \{c_{1,2,3}^h\}} F_1^h(c) + \sum_{c_0, c_1 \in \{c_{1,2,3}^h\}} F_2^h(c_0, c_1), \quad B^h = \sum_{c_0 \in \{c_{1,2,3}^h\}, c_y \in Y} F_2^h(c_0, c_y), \quad \text{and} \quad C^h = \sum_{c_1 \in \{c_{1,2,3}^h\}, c_y \in Y} F_2^h(c_y, c_1). \quad (16)$$

eq:ABC

We claim that $A^l = A^r$, $B^l = B^r$, and $C^l = C^r$.

To show that $A^l = A^r$, we need to compare polynomials in $g_{\nu\alpha\beta}^l$ with polynomials in $g_{\nu\alpha\beta}^r$ in which α and β may belong to the set $\{i^+, j^+, k^+\}$ on which it may be that $g^l \neq g^r$. Fortunately the g -rules of Equations (7) and (8) allow us to rewrite the offending g 's, namely the ones with subscripts in $\{i^+, j^+, k^+\}$, in terms of other g 's whose subscripts are in $\{i, j, k, i^{++}, j^{++}, k^{++}\}$, where $g^l = g^r$. So it is enough to show that

$$A^l /. (\text{the } g\text{-rules for } c_1^l, c_2^l, c_3^l) = A^r /. (\text{the } g\text{-rules for } c_1^r, c_2^r, c_3^r) \quad \text{under } g^l = g^r, \quad (17)$$

eq:R3A

where the symbol $/.$ means “apply the rules”. This is a finite computation that can in-principle be carried out by hand. But each A^h is a sum of $3 + 9 = 12$ polynomials in the g^h 's, these polynomials are rather unpleasant (see (3) and (4)), and applying the relevant g -rules adds a bit further to the complexity. Luckily, we can delegate this pages-long calculation to an entity that doesn't complain.

First, we implement the Kronecker δ -function, the g -rules for a crossing (s, i, j) , and the g -rules for a list of crossings X :

```

(☹)  $\delta_{i,j} := \text{If}[i === j, 1, 0];$ 
(♥)  $\text{gRules}[\{s, i, j\}] := \{$ 
     $\text{g}_{v,j\beta} \Rightarrow \text{g}_{vj+\beta} + \delta_{j\beta}, \text{g}_{v,i\beta} \Rightarrow \text{T}_v^s \text{g}_{vi+\beta} + (1 - \text{T}_v^s) \text{g}_{vj+\beta} + \delta_{i\beta},$ 
     $\text{g}_{v,\alpha i^+} \Rightarrow \text{T}_v^s \text{g}_{v\alpha i} + \delta_{\alpha i^+}, \text{g}_{v,\alpha j^+} \Rightarrow \text{g}_{v\alpha j} + (1 - \text{T}_v^s) \text{g}_{v\alpha i} + \delta_{\alpha j^+}$ 
 $\};$ 
 $\text{gRules}[X\_List] := \text{Union} @@ \text{Table}[\text{gRules}[c], \{c, \{X\}\}]$ 


```

We then let $X1$ be the three crossings in the left-hand-side of the R3 move, as in Figure 4.3, we let $A1$ be the A^l term of (16), and we let lhs be the result of applying the g -rules for the

crossings in X_1 to A_1 . We print only a “Short” version of `lhs` because the full thing would cover about 2.5 pages:

```

☹ X1 = {{1, j, k}, {1, i, k+}, {1, i+, j+}};
♥ A1 = Sum[F1[c], {c, X1}] + Sum[F2[c0, c1], {c0, X1}, {c1, X1}];
lhs = Simplify[A1 //. gRules @@ X1];
Short[lhs, 5]

 
$$-\frac{1}{2(1-T_2)} \left( 3 - 3T_2 + \ll 129 \gg + \right.$$


$$2(1-T_2) \left( 1 + T_2 (T_2 g_{2, \ll 1 \gg^+, i} - (-1 + T_2) g_{2, \ll 1 \gg, i}) - (-1 + T_2) g_{2, (k^+)^+, i} \right)$$


$$\left. \left( 1 + (1 - T_1 T_2) g_{3, (k^+)^+, j} + g_{3, (k^+)^+, k} \right) \right)$$


```

We do the same for A^r , except this time, without printing at all:

```

☹ Xr = {{1, i, j}, {1, i+, k}, {1, j+, k+}};
♥ Ar = Sum[F1[c], {c, Xr}] + Sum[F2[c0, c1], {c0, Xr}, {c1, Xr}];
rhs = Simplify[Ar //. gRules @@ Xr];

```

We then compare `lhs` with `rhs`. The output, `True`, tells us that we have proven (17):


```

☹ Simplify[lhs == rhs]  True

```

We show that $B^l = B^r$ by following exactly the same procedure. Note that we ignore the summation over c_y and instead treat it as a fixed crossing $c_y = (s, m, n)$. If an equality is proven for every fixed c_y , it is of course also proven for the sum over $c_y \in Y$.


```

☹ lhs = Sum[F2[c0, {s, m, n}], {c0, X1}] //. gRules @@ X1;  True
♥ rhs = Sum[F2[c0, {s, m, n}], {c0, Xr}] //. gRules @@ Xr;
Simplify[lhs == rhs]

```

Similarly we prove that $C^l = C^r$, and this concludes the proof of Proposition 7.

```

☹ lhs = Sum[F2[{s, m, n}, c1], {c1, X1}] //. gRules @@ X1;  True
♥ rhs = Sum[F2[{s, m, n}, c1], {c1, Xr}] //. gRules @@ Xr;
Simplify[lhs == rhs] □

```

rem:E

Remark 8. The computations above were carried out for generic $g_{\nu\alpha\beta}$ and for a generic $c_y = (s, m, n)$; namely, without specifying the knot diagrams in full, and hence without assigning specific values to $g_{\nu\alpha\beta}$, and without specifying m and n . Under these conditions the three parts of (16) cannot mix (namely, terms from, say, A^h cannot cancel terms in B^h or C^h), and so it would have been enough to show that $E^l = E^r$, where E^h combines A^h and B^h and C^h (and a few harmless further terms) by adding c_y to the summation corresponding to A^h :

$$E^h = \sum_{c \in \{c_{1,2,3,y}^h\}} F_1^h(c) + \sum_{c_0, c_1 \in \{c_{1,2,3,y}^h\}} F_2^h(c_0, c_1).$$

But that’s a simpler computation:

```

☹ ESum[X_] := (Sum[F1[c], {c, X}] + Sum[F2[c0, c1], {c0, X}, {c1, X}]) //. gRules @@ X;

```

☹ $Xl = \{\{1, j, k\}, \{1, i, k^+\}, \{1, i^+, j^+\}\};$

❤ $Xr = \{\{1, i, j\}, \{1, i^+, k\}, \{1, j^+, k^+\}\};$


$\text{Simplify}[\text{ESum}[\text{Append}[Xl, \{s, m, n\}]] = \text{ESum}[\text{Append}[Xr, \{s, m, n\}]]]$

True

prop:R2c

Proposition 9. *The quantity θ is invariant under $R2c^+$.*

Proof. We follow the same logic as in the proof of Proposition 7, as simplified by Remark 8. We start with the figure that replaces Figure 4.3 (note the null vertices in D^r and their minimal effect as in Lemma 4 and Remark 5):



To compute “ E ” sums as in Remark 8 we first have to extend the **ESum** routine to accept also a list R of pairs (φ, k) of the form (rotation number, edge label):

☹ $\text{ESum}[X_, R_] :=$

❤ $(\text{Sum}[\text{F}_1[\mathbf{c}], \{\mathbf{c}, X\}] + \text{Sum}[\text{F}_2[\mathbf{c0}, \mathbf{c1}], \{\mathbf{c0}, X\}, \{\mathbf{c1}, X\}] + \text{Sum}[\text{F}_3@@\mathbf{r}, \{\mathbf{r}, R\}]) // \text{gRules}@@X;$

We then compute E^l by calling **ESum** with crossings $(-1, i, j^+)$, $(1, i^+, j)$ as in the left hand side of the $R2c^+$ moves, a generic extra crossing (s, m, n) , and a rotation number of 1 on edge j^+ :

☹ $E1 = \text{Simplify}[\text{ESum}[\{\{-1, i, j^+\}, \{1, i^+, j\}, \{s, m, n\}\}, \{\{1, j^+\}\}]];$

Short[lhs, 5]

$$\frac{1}{2(-1 + T_2^5)} \left((1 + s + 2s(T_1 T_2)^5 g_{3,m^+,m} + \ll 11 \gg + 2g_{3,(j^+)^+,j^-} \right. \\ \left. T_2^5 (1 + s - 2s g_{1,n^+,m} g_{2,n^+,m} + 2s g_{2,n^+,n} + \ll 28 \gg + 2s g_{2,m^+,m} (1 + g_{3,n^+,n}) + 2g_{3,(j^+)^+,j}) \right)$$

The computation of E^r is simpler, as it only involves the generic (s, m, n) and the rotation $(1, j^+)$. We apply the g -rules of Equation (10) “by hand” on $g_{\nu\alpha\beta}$ (only if $\alpha \in \{i, i^+, j, j^+\}$) and then compare E^l with E^r to conclude the proof:

☹ $E_r = \text{Simplify}[\text{ESum}[\{\{s, m, n\}\}, \{\{1, j\}\}]] //$

❤ $g_{\nu, \alpha, \beta} // \text{MemberQ}[\{i, i^+, j, j^+\}, \alpha] \Rightarrow \delta_{\alpha, \beta} + g_{\nu, \alpha^+, \beta};$

$\text{Simplify}[E1 == Er]$

True

□

5. STRONG AND MEANINGFUL

MORE. Do not forget to say something about volumes.

6. CONJECTURES AND DREAMS

REFERENCES

- [Al] J. W. Alexander, *Topological invariants of knots and link*, Trans. Amer. Math. Soc. **30** (1928) 275–306. See pp. 3.
- [BN1] D. Bar-Natan, *Cars, Interchanges, Traffic Counters, and some Pretty Darned Good Knot Invariants*, talk given in Oaxaca, October 2022. Video and handout at <http://drorbn.net/oa22>. See pp. 16, 17.
- [BN2] Dror Bar-Natan, *Knot Invariants from Finite Dimensional Integration*, talks in Beijing (July 2024, <http://drorbn.net/icbs24>) and in Geneva (August 2024, <http://drorbn.net/ge24>). See pp. 8.
- [BN3] Dror Bar-Natan, *The Strongest Genuinely Computable Knot Invariant in 2024*, talk given in Toronto (October 2024, <http://drorbn.net/to24>). See pp. 8, 16, 17.
- [BV1] D. Bar-Natan and R. van der Veen, *A Perturbed-Alexander Invariant*, Quantum Topology **15** (2024) 449–472, [arXiv:2206.12298](https://arxiv.org/abs/2206.12298). See pp. 7, 8, 15, 16, 17, 18.
- [BV2] D. Bar-Natan and R. van der Veen, *A Very Fast, Very Strong, Topologically Meaningful and Fun Knot Invariant*, (self-reference), paper and related files at <http://drorbn.net/Theta>. The [arXiv:2206.12298](https://arxiv.org/abs/2206.12298) edition may be older. See pp. 11.
- [DHOEBL] N. Dunfield, A. Hirani, M. Obeidin, A. Ehrenberg, S. Bhattacharyya, D. Lei, and others, *Random Knots: A Preliminary Report*, lecture notes at https://nmd.web.illinois.edu/slides/random_knots.pdf. Also a data file at https://drorbn.net/AcademicPensieve/People/Dunfield/nmd_random_knots. See pp. 5.
- [GR] S. Garoufalidis and L. Rozansky, *The Loop Expansion of the Kontsevich Integral, the Null-Move, and S-Equivalence*, [arXiv:math.GT/0003187](https://arxiv.org/abs/math.GT/0003187). See pp. 1.
- [Kr] A. Kriker, *The Lines of the Kontsevich Integral and Rozansky’s Rationality Conjecture*, [arXiv:math/0005284](https://arxiv.org/abs/math/0005284). See pp. 1.
- [Oh] T. Ohtsuki, *On the 2-Loop Polynomial of Knots*, Geometry & Topology **11** (2007) 1357–1475. See pp. 1.
- [Ro1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten’s Invariant of 3D Manifolds, I*, Comm. Math. Phys. **175-2** (1996) 275–296, [arXiv:hep-th/9401061](https://arxiv.org/abs/hep-th/9401061). See pp. 1.
- [Ro2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial*, Adv. Math. **134-1** (1998) 1–31, [arXiv:q-alg/9604005](https://arxiv.org/abs/q-alg/9604005). See pp. 1.
- [Ro3] L. Rozansky, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture*, [arXiv:math/0201139](https://arxiv.org/abs/math/0201139). See pp. 1.
- [Wo] *Wolfram Language & System Documentation Center*, <https://reference.wolfram.com/language/>. See pp. 11.

DEPARTMENT OF MATHEMATICS, UNIVERSITY OF TORONTO, TORONTO ONTARIO M5S 2E4, CANADA
Email address: drorbn@math.toronto.edu
URL: <http://www.math.toronto.edu/drorbn>

UNIVERSITY OF GRONINGEN, BERNOULLI INSTITUTE, P.O. BOX 407, 9700 AK GRONINGEN, THE NETHERLANDS
Email address: roland.mathematics@gmail.com
URL: <http://www.rolandvdv.nl/>